
РЕШЕНИЕ ПРОСТЕЙШЕЙ ЗАДАЧИ ВАРИАЦИОННОГО 

ИСЧИСЛЕНИЯ КОНЕЧНО-РАЗНОСТНЫМ МЕТОДОМ ЭЙЛЕРА 

 В ПАКЕТЕ SCILAB 

 

 Простейшая задача вариационного исчисления сводится к решению 

краевой задачи для уравнения Эйлера, которое в общем случае представляет 

собой дифференциальное уравнение второго порядка. Аналитически такая 

задача решается для ограниченного числа уравнений, поэтому в общем 

случае применяются приближенные методы решения. Одним из таких 

методов является конечно-разностный метод Эйлера, который заключается в 

том, что дифференциальное уравнение приводится к системе уравнений 

относительно значений искомой функции в точках разбиения. 

 В данной работе представлено графическое приложение в пакете Scilab 

(www.scilab.org.), позволяющее находить приближенной решение 

простейшей вариационной задачи для некоторого класса подынтегральных 

функций. 

 Пусть требуется найти решение простейшей вариационной задачи 

следующего вида 

     
 

 
2 2

1 2 3[ ] ,

b
a

ba

y a y
V y f x y f x yy f x y dx extr

y b y

 
        
 ,(1) 

http://www.scilab.org/

где  y x  – непрерывно дифференцируемая функция на отрезке  ;a b . 

 Составим вычислительную задачу для задачи (1) в соответствии с 

алгоритмом конечно-разностного метода Эйлера. 

 Для решения простейшей вариационной задачи 

     0[ ] , , , ,

b

n

a

V y F x y y dx y a y y b y     (2) 

конечно-разностным методом Эйлера заменяем в (2) функцию  y x  ломаной 

с вершинами в точках с равноотстоящими абсциссами 1i ix x x    . 

         0 0 1 1 2 2 1 1, , , , , ,..., , , ,n n n nx a y x y x y x y x b y   . 

Производную в задаче (2) заменяем разностным отношением 1i i
i

y y
y

x

 
 


, а 

интеграл заменяем суммой по формуле прямоугольников 

 .  (3) 

Необходимое условие экстремума – равенство нулю частных производных. 

Заметим, что iy  входит во второй и третий аргументы подынтегральной 

функции в (3) 

В нашем случае имеем 

    (4) 

 Таким образом, получаем систему (4) линейных уравнений 

относительно значений iy  искомой функции. Учитывая, что 0y  и ny  заданы, 

эта система содержит 1n   уравнений относительно 1n   неизвестных 

1 2 3 1, , , ..., ny y y y  . 

 Система (4) является линейной системой, поэтому она может быть 

решена встроенной функцией Scilab для решения линейных систем. После 

того как решение найдено, необходимо определить экстремальное значение 

функционала. Вычисление этого значения выполним, заменяя значения 

аргумента x , функции y  и ее производной y  значениями в средних точках 

отрезков разбиения 

1 1 1, ,

Запишем программу в Scilab, реализующую графическое приложение для 

решения поставленной простейшей вариационной задачи 

Листинг программы в Scilab 

// Задание формы 

F=figure('figure_name','Пример','position',[20,200,1200,500],.. 

'BackgroundColor',[0.9,0.9,1]); 

//Задание условия задачи 

t1=uicontrol(F,"style","text",'position',[50,440,600,40],.. 

'string','Решить простейшую вариационную задачу конечно-разностным 

методом Эйлера',.. 

'FontName','Times New Roman','FontSize',16,.. 

'FontWeight','bold','FontAngle','italic','BackgroundColor',[0.9,0.9,1]); 

// Задание вида уравнения 

t2=uicontrol(F,"style","text",'position',[50,380,600,60],.. 

'string','$\it V[y]=\int_a^b(f_1(x)y^2+f_2(x)yy+f_3(x)y''^2)dx \to min, y(a)=y_a, 

y(b)=y_b$',.. 

'BackgroundColor',[0.9,0.9,1],'FontSize',14); 

// Задание начала и конца отрезка 

lab_a=uicontrol(F,'style','text','string','a=','position',[150,350,20,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_a=uicontrol(F,'style','edit','string',' ','position',[170,350,50,20]); 

lab_b=uicontrol(F,'style','text','string','b=','position',[230,350,20,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_b=uicontrol(F,'style','edit','string',' ','position',[250,350,50,20]); 

// Задание числа отрезков разбиения 

lab_n=uicontrol(F,'style','text','string','n=','position',[310,350,20,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_n=uicontrol(F,'style','edit','string',' ','position',[330,350,50,20]); 

//Задание краевых условий 

lab_ya=uicontrol(F,'style','text','string','$y_a=$','position',[390,350,30,20],.. 



71 
 

'BackgroundColor',[0.9,0.9,1]); 

edit_ya=uicontrol(F,'style','edit','string',' ','position',[420,350,50,20]); 

lab_yb=uicontrol(F,'style','text','string','$y_b=$','position',[480,350,30,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_yb=uicontrol(F,'style','edit','string',' ','position',[510,350,50,20]); 

// Задание функций 

lab_f1=uicontrol(F,'style','text','string','$f_1(x)= $','position',[160,300,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_f1=uicontrol(F,'style','edit','string','function y=f1(x),y= ,endfunction',.. 

'position',[220,300,300,25]); 

lab_f2=uicontrol(F,'style','text','string','$f_2(x)= $','position',[160,270,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_f2=uicontrol(F,'style','edit','string','function y=f2(x),y= ,endfunction',.. 

'position',[220,270,300,25]); 

lab_f3=uicontrol(F,'style','text','string','$f_3(x)= $','position',[160,240,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

edit_f3=uicontrol(F,'style','edit','string','function y=f3(x),y= ,endfunction',.. 

'position',[220,240,300,25]); 

//Задание управляющих кнопок 

button_1=uicontrol(F,'style','pushbutton','string',.. 

'Получить решение вариационной задачи','position',[150,150,300,25],.. 

'callback','solve_tab'); 

//Вывод результатов 

lab_x=uicontrol(F,'style','text','string','$x= $','position',[70,100,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

lab_y=uicontrol(F,'style','text','string','$y= $','position',[70,80,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

lab_V=uicontrol(F,'style','text','string','$V= $','position',[70,60,40,20],.. 

'BackgroundColor',[0.9,0.9,1]); 

result_1=uicontrol(F,'style','text','string',' ','position',[100,100,450,20]); 

result_2=uicontrol(F,'style','text','string',' ','position',[100,80,450,20]); 

result_3=uicontrol(F,'style','text','string',' ','position',[100,60,450,20]); 

// Функции реализации алгоритма 

function solve_tab() 

    a=eval(get(edit_a,'string')); 

    b=eval(get(edit_b,'string')); 

    n=eval(get(edit_n,'string')); 

    h=(b-a)/n; 

    x0=a; 

    xn=b; 

    y0=eval(get(edit_ya,'string')); 

    yn=eval(get(edit_yb,'string')); 

    execstr(get(edit_f1,'string')); 

    execstr(get(edit_f2,'string')); 

    execstr(get(edit_f3,'string')); 

    x=(a+h):h:b; 

    C=zeros(n-1,n-1); 

    C(1,1)=2*(f1(x(1))*h+f2(x(1))+(f3(x(1))+f3(x(2)))/h); 

    C(1,2)=-f2(x(2))-2*f3(x(2))/h; 

    C(n-1,n-2)=-f2(x(n-1))-2*f3(x(n-1))/h; 

    C(n-1,n-1)=2*(f1(x(n-1))*h+f2(x(n-1))+(f3(x(n-1))+f3(x(n)))/h); 

    for k=2:n-2 

      C(k,k-1)=-f2(x(k))-2*f3(x(k))/h; 

      C(k,k)=2*(f1(x(k))*h+f2(x(k))+(f3(x(k))+f3(x(k+1)))/h); 

      C(k,k+1)=-f2(x(k+1))-2*f3(x(k+1))/h; 

    end 

    B=zeros(n-1,1); 

    B(1,1)=-(f2(x(1))+2*f3(x(1))/h)*y0; 

    B(n-1,1)=-(f2(x(n))+2*f3(x(n))/h)*yn; 

    format('v',6); 

    y=linsolve(C,B); 

    y=[y0,y',yn]; 

    x=[a,x] 

    for k=1:n 

       u(k)=(x(k)+x(k+1))/2; 

       v(k)=(y(k)+y(k+1))/2; 

       w(k)=(y(k+1)-y(k))/h; 

    end 

   V=0; 

   for k=1:n 

       V=V+(f1(u(k))*(v(k))^2+f2(u(k))*v(k)*w(k)+f3(u(k))*(w(k))^2)*h; 

   end 

    z1=sci2exp(x); 

    z2=sci2exp(y); 

    z3=sci2exp(V); 

    set(result_1,'string',z1); 

    set(result_2,'string',z2); 

    set(result_3,'string',z3); 

    subplot(1,2,2);  

    plot(x,y); 

endfunction 

Выполнение этой программы приводит к появлению окна графического 

приложения. На рисунке 1 представлена часть этого графического окна, где 

вводятся данные задачи и выводится результат. 

В качестве примера использования созданного графического прило-

жения рассмотрим решение следующей вариационной задачи (рисунок 2) 

     
1

2 2

0

[ ] , 0 0, 1 1V y y y dx y y   

Рисунок 1 – Часть окна графического приложения для ввода данных 

 

Рисунок 2 – Пример решения вариационной задачи 

 Отметим, что для проверки достаточного условия существования 

экстремума используется условие Лежандра, которое в данной задаче 

определяется знаком функции  32y yF f x   . Так как в данном примере 

2 0y yF     , то согласно условию Лежандра, экстремаль для этого примера 

дает минимум функционалу. 

 Данная задача имеет аналитическое решение. Уравнение Эйлера для 

нее запишется в виде,  

  2 2, , , 2 , 2 ,

0 2 2 0, 0.

y y

y y

F x y y y y F y F y

d
F F y y y y

dx





     

       
 

 Экстремали вариационной задачи определяются из решения уравнения 

0y y   : 1 2

x xy C e C e  . 

Далее, подставляем краевые условия 

 

   

1 2
1 21 2

11

21 2
2 2

,,0 0, 0, 1

1 1. 1.1.
.

1

e
CC Cy C C e

y C e eC e C e e
C

e




        

     
       

 

 

Экстремалью, удовлетворяющей краевым условиям, будет функция  

2 2 2 2

2 2
sh

21 1 1 1

x x
x xe e e e e e

y e e y y x
e e e e


 

     
   

. 

Расчет значения функционала для этой экстремали дает 

2 21

2 2
0

2 2
[ ] sh ch

1 1

e e
V y x x dx

e e

    
            
  

 
 

   

1

1 12 2 2
2 2

2 2 2
2 2 2

0 0
0

4 4 2
sh ch ch 2 sh 2 1,313.

1 1 1

e e e
x x dx xdx x

e e e

    

  
   

Таким образом, значение функционала, рассчитанное с помощью 

графического приложения приближенным конечно-разностным методом 

Эйлера и значение функционала, найденное аналитически, практически 

совпадают. 

